NOAA Western Region Computer Programs and
Problems NWS WRCP - No. 39

850-MILLIBAR CHARTS DERIVED FROM SURFACE DATA

Salt Lake City, Utah
December 1982

U.S. DEPARTMENT OF COMMERCE / National Oceanic and Atmospheric Administration / National Weather Service

PREFACE

This Western Region publication series is considered as a subset of our Technical Memorandum series. This series will be devoted exclusively to the exchange of information on and documentation of computer programs and related subjects. This series was initiated because it did not seem appropriate to publish computer program papers as Technical Memoranda; yet, we wanted to share this type of information with all Western Region forecasters in a systematic way. Another reason was our concern that in the developing AFOS-era there will be unnecessary and wasteful duplication of effort in writing computer programs in National Weather Service (NWS). Documentation and exchange of ideas and programs envisioned in this series hopefully will reduce such duplication. We also believe that by publishing the programming work of our forecasters, we will stimulate others to use these programs or develop their own programs to take advantage of the computing capabilities AFOS makes available.

We solicit computer-oriented papers and computer programs from forecasters for us to publish in this series. Simple and short programs should not be prejudged as unsuitable.

The great potential of the AFOS-era is strongly related to local computer facilities permitting meteorologists to practice in a more scientific environment. It is our hope that this new series will help in developing this potential into reality.

NOAA WESTERN REGION COMPUTER PROGRAMS AND PROBLEMS NWS WRCP

1 Standardized Format for Computer Series.
2 AFOS Crop and Soil Information Report Programs. Ken Mielke, July 1979.
3 Decoder for Significant Level Transmissions of Raobs. John A. Jannuzzi, August 1979.
4 Precipitable Water Estimate. Elizabeth Morse, October 1979.
5 Utah Recreational Temperature Program. Kenneth M. Labas, November 1979.
6 Normal Maximum/Minimum Temperature Program for Montana, Kenneth Mielke, December 1979.
7 Plotting of Ocean Wave Energy Spectral Data. John R. Zimmerman, December 1979.
8 Raob Plot and Analysis Routines. John A. Jannuzzi, January 1980.
9 The SWAB Program. Morris S. Webb, Jr., April 1980. (PB80-196041)
10 Flash-Flood Procedure. Donald P. Laurine and Ralph C. Hatch, April 1980. (PB80-298658)
11 Program to Forecast Probability of Summer Stratus in Seattle Using the Durst Objective Method. John R. Zimmerman, May 1980.
12 Probability of Sequences of Wet and Dry Days. Hazen H. Bedke, June 1980. (PB80-223340)
13 Automated Montana Hourly Weather Roundup. Joe L. Johnston, July 1980. (PB81-102576)
14 Lightning Activity Levels. Mark A. Mollner, July 1980. (PB81-108300)
15 Two Fortran Applications of Wind-Driven Ekman Water Transport Theory: Upwelling Index and Storm Tide. Kent S. Short, July 1980. (PB81-102568)
16 AFOS System Local Data Base Save and Rebuild Procedures or A Master Doomsday Program. Brian W. Finke, July 1980. (PB81-108342)
17 AFOS/RDOS Translator Subroutine. Morris S. Webb, Jr., August 1980. (PB81-108334)
18 AFOS Graphics Creation from Fortran. Alexander E. MacDonald, August 1980. (PB81-205304)
19 DATAKEYØ Repair Program. Paul D. Tolleson, August 1980. (PB81-102543)
20 Contiguous File Transfer from the DPCM to the DCM. Paul D. Tolleson, September 1980. (PB81-128035)
21 Freezing Level Program. Kenneth B. Mielke, September 1980. (PB81-128043)
22 Radar Boresighting Verification Program. Thomas E. Adler, November 1980.
23 Accessing the AFOS Data Base. Matthew Peroutka, January 1981.
24 AFOS Work Processor. Morris S. Webb, Jr., February 1981. (PB81-210007)
25 Automated Weather Log for Terminal Forecasting. John A. Jannuzzi, February 1981. (PB81-210999)
26 Program to Computer Downwind Concentrations from a Toxic Spill. John R. Zimmerman, February 1981. (PB81-205296)
27 Animation of AFOS Graphics. Joe Wakefield and Jim Fors, April 1981.
28 AFOS Interactive Graphics. Jim Fors, Don Laurine, and Sandy MacDonald, April 1981.
29 Computer Programs for Aviation Forecast Transmission. Kenneth B. Mielke and Matthew R. Peroutka, May 1981.
30 AFOS Product Collective Program. Morris S. Webb, Jr., September 1981.
31 Graphic Display of FOUS Output. Stephen D. Steenrod, September 1981.
32 Automation of Hourly Aviation Observation Calculations. W. Paul Duval, October 1981.
33 Mesoscale Objective Analysis. Andrew J. Spry and Jeffrey L. Anderson, December 1981.
34 Orographic Snowfall Rate Model for Alta, Utah. Steven K. Todd and Glenn E. Rasch, December 1981.
35 F-6 Monthly Climatic Summary Program for AFOS. Peter G. Mueller, May 1982.

NOAA Western Region Computer Programs and Problems NWS WRCP - No. 39

850-MILLIBAR CHARTS DERIVED FROM SURFACE DATA

Jeffrey L. Anderson
National Weather Service
Western Region Headquarters
Scientific Services Division

December 1982

UNITED STATES DEPARTMENT OF COMMERCE
Malcolm Baldrige, Secretary / National Oceanic and Atmospheric Administration
John V. Byrne, Administrator / National Weather Service
Richard E. Hallgren, Director

This technical publication has been reviewed and is approved for publication by Scientific Services Division, Western Region.

Glenn E. Rasch

Glenn E. Rasch, Acting Chief
Scientific Services Division
Western Region Headquarters
Salt Lake City, Utah

CONTENTS

FIGURES

Page

850-MILLIBAR CHARTS DERIVED FROM SURFACE DATA

Jeffrey L. Anderson*

I. General Information

A. Summary:

Representative analyses of data near the earth's surface in regions of high and rugged terrain are notoriously difficult to produce. Standard surface analyses are often highly inaccurate in areas, such as the western United States, where there is a great difference in elevation between actual surface reporting stations and the hypothetical subterranean "surface" to which observations are reduced. Reducing observations to some level closer to the mean elevation of the actual earth's surface can produce much more representative analyses. In the western United States, the 850-mb surface is very close to most surface reporting stations; analyses reduced to 850 mb are often more accurate than normal surface analyses in this region. AFOS programs have been written to reduce surface aviation observations to the 850-mb surface. 850-mb height, potential temperature, and geostrophic vorticity fields can be produced.

The three 850-mb fields can be produced hourly from surface aviation observation. The 850-mb height field can be used as an equivalent of a surface pressure chart. The 850-mb potential temperature chart eliminates many of the local effects that bias normal surface temperature charts, and is therefore an excellent indicator of differences in air masses. Hence, surface fronts can be placed along areas of strong gradients of potential temperature. 850-mb geostrophic vorticity is also a good indicator of frontal position. Fronts tend to be located just ahead of areas of strong positive vorticity.

The 850 charts are produced by programs that reduce surface data to 850 mb. The data are then analyzed by portions of the Western Region mesoscale analysis programs. Graphics are produced on a fixed location grid covering the higher potions of the western United States.

B. Environment:

The programs, both written in Data General Fortran IV, are designed to run in the background of AFOS. Both programs are actually additions to the Western Region mesoscale objective analysis series. The first new program is chained to by MESO1, the first program in the mesoscale, and acts as an alternate second program in the mesoscale series. The second new program executes after MESO4 but before MESO5 in the mesoscale series.

C. References:

Bullock, Carl: "A Derived 850 Millibar Chart for the Western U.S. and Some Diagnostic Uses". University of Wisconsin Master Thesis, 1978.

*Summer Student Trainee, Western Region Headquarters, Scientific Services Division.

Spry, Andrew J. and Jeffrey L. Anderson: "Mesoscale Objective Analysis", NWS WRCP No. 33, December, 1981.

II. Application

A. Program Description:

The 850-mb reduction programs are designed to give the forecaster access to an hourly 850-mb chart which better depicts surface synoptic scale features than do normal surface charts. The 850-mb potential temperature and geostrophic vorticity charts are not available from any other source and are useful in locating surface fronts.

The 850-mb reduction series is run as part of the mesoscale analysis series. The programs obtain graphics output names, contour intervals, and smoothing passes from the same menu preformat that is used by the regular mesoscale. Unlike the regular mesoscale, both the height and temperature fields will always be run while only the vorticity field is optional. The program series begins by reading in surface data. The first new program, MESO2E, reduces the data to the 850-mb surface. The 850-mb data is then gridded and analyzed by MESO3 and MESO4 of the original mesoscale. If an 850-mb vorticity field is requested, geostrophic wind components are computed by the second new program, MESO4E. These components are then used to compute vorticity in original mesoscale program MESO5.

The surface aviation observations are reduced to 850 mb using equations developed by Bullock. The following equations summarize the approach used in the subroutine H850 of MESO2E to compute both the 850-mb heights and potential temperature.

1. Vapor pressure is computed.

$$e = 6.11\text{mb} * 10^{aT_D/(T_D + b)}$$

T_D = dewpoint in ^{o}C
e = vapor pressure
a = 7.5
b = 237.3 C

2. Second, a correction is applied to the surface temperature to reduce diurnal effects. This correction is not applied between 04Z and 16Z if the surface wind speed is greater than 14 kts.

$$T_s = T - 2.9°C \sin [2\pi/24 * (t-16)]$$

where
T_s = adjusted temperature in ^{o}C
T = observed surface temperature in ^{o}C
t = hour of observation in GMT hours

3. Next, true surface pressure is computed from altimeter setting and station elevation (pressure shown on a standard surface map has already been reduced to sea level).

$$P_s = (A^n - P_\emptyset^n \; a\, H/T_\emptyset)^{1/n}$$

with
A = altimeter setting in inches of mercury
$P_\emptyset$ = 29.921 inches of mercury
$T_\emptyset$ = 288.15 K
a = .0065 K/m
H = station elevation in meters
n = 0.1902632

4. Finally, 850-mb height is computed.

$$\Phi_{850} = \Phi_s - [\log_{10}(850/P_s) * (T_s + ec)] / [K-(a/2) * \log_{10}(850/P_s)]$$

where
Φ_{850} = 850-mb height in meters
Φ_s = surface height in meters
P_s = surface pressure in mb
T_s = diurnally corrected surface temperature
e = vapor pressure
c = 0.12°C/mb
K = 0.0148275 °C/m
a = 5 °C/km

5. 850-mb temperature is calculated using an assumed lapse rate of 5 °C/km

$$T_{850} = T_s + (\Phi_{850} - \Phi_s) * 5°C/Km$$

all symbols are defined above.

6. Last, 850-mb potential temperature is computed using Poisson's equation

$$\theta_{850} = T_{850} \; (1000/850)^{R/C_p}$$

Once the geopotential height field exists, it can be used to compute the u and v components of the geostrophic wind using the relations $u = -1/f \; d\Phi/dy$ and $v = 1/f \; d\Phi/dx$. These in turn are used to compute vorticity.

B. Machine Requirements:

All new or modified mesoscale programs require less than 17K of memory to execute. The new programs, MESO2E and MESO4E, require respectively 62 and 34 blocks of disk storage. Neither new program requires an overlay and both require a maximum of four channels. Runtime for the entire 850-mb reduction series is highly variable depending on the number of surface aviation observations available, but an average time is just over three minutes.

C. Structure of Software:

The interaction of the original mesoscale analysis programs and the new 850-mb reduction programs is rather complex and is illustrated in Figure 1.

The following modifications were made to the original mesoscale programs in order to incorporate the 850-mb programs:

1. MESO1 -- An input type 4, "850-mb Saodata", was added to the mesoanalysis menu preformat to trigger the 850-mb series. A new subroutine, MESO850, is called to process the MANALINFO file needed for the 850-mb series. Statements were added to chain to MESO2E if an 850-mb reduction chart is requested on the menu.

2. MESO4 -- A statement was added to check the 43rd element of MANALINFO, also called DELFLG, for the value 850. If the value is 850, the control is chained to MESO4E to begin production of a vorticity field.

The two new programs written for the 850-mb series execute as follows:

1. MESO2E -- Reads in the MANALINFO file created in MESO1. The base longitude and latitude for the requested map background are computed and subroutine GETSAO is called. Upon return, a MANALOBS file is created and control is passed to MESO3 in the main mesoscale.

   a. GETSAO -- Reads the altimeter setting, temperature, dewpoint, and wind velocity fields from the SAODATA file. For each station, elevation, latitude, and longitude are obtained from the STDIR.MS file via subroutine BNSCH. Stations with elevations less than 700 or greater than 2060 meters are discarded to prevent errors from large reductions. Subroutine H850 is called to compute 850-mb geopotential height and potential temperature. Subroutine BOGUS is called to create a new MANALINFO file to fool the main mesoscale series into processing the 850-mb data.

2. MESO4E -- Reads in the gridded height data file produced by MESO3 and computes the u and v wind components. After being smoothed twice by subroutine SMOOTH, the wind components are written to files and a MANALINFO file is created to fool MESO5 into producing 850-mb vorticity. Control is then chained to MESO5.

The load lines for the new programs are:

1. MESO2E -- RLDR MESO2E GETSAO H850 BOGUS BNSCH BG.LB UTIL.LB FORT.LB

2. MESO4E -- RLDR MESO4E SMOOTH UTIL.LB FORT.LB



FIGURE 1. THE INTERRELATIONS OF THE MESOSCALE ANALYSIS PROGRAMS. PROGRAM UNITS ARE INDICATED BY BOXES AND DATA FILES BY OVALS. PROGRAM CHAINS ARE INDICATED BY CHAINS WHILE LINES INDICATE FILE USE.

D. Data Base

An SAODATA file, produced by the SAO decoder, is needed to execute the 850-mb series. Data files are used to pass relevant information between program units in the mesoscale analysis series. MESO2E accesses the files SAODATA and MANALINFO. A new MANALINFO file is created along with a MANALOBS file containing latitude, longitude, potential temperature, and 850-mb heights. MESO4E reads in a NAMES file, MANALINFO, and a gridded 850-mb height data file. NAMES and MANALINFO are modified to pass information to MESO5 and gridded u and v wind component data files are created. All file formats are documented in the original mesoscale CP #33.

III. Procedures

A. Initiation of Program:

The command RUN:MESO1 from an AFOS ADM will execute the program series.

B. Input Required:

To run the 850-mb reduction series, the mesoanalysis preformat used for the original mesoscale must be filled out to specify which fields are desired. The desired number of smoothing passes, the contour interval, and the output graphic name may also be specified. Both temperature and height fields must be produced each time the series is run, so requesting just one of these fields will result in the production of the other using default parameters. The user does have complete control over production of the vorticity field provided that one of the other fields (temperature or height) is requested. The user has no control over the geographical location of the analysis grid since the 850-mb reduction is only valid in regions of high elevation. Either North American or U.S. map background may be selected in the preformat. A sample preformat menu with recommended values for producing heights, potential temperature, and geostrophic vorticity, is shown in Figure 2.

```
                          ANALYSIS MENU
      (   ) REFER TO UPPER AIR PLOTS-------? MEANS NOT USER SELECTABLE
FIELD           UNITS     AVAILABLE ANALYZE SMOOTHING CONTOUR  GRAPHIC NAME
                            FROM              PASSES  INTERVAL   NMCGPH???
TEMPERATURE     DEG F(C)   0-1-2-3-4  [X]       [2]      [   2](2)  [TMP]
PRES(HEIGHT)    MB(METERS) 0-1-2-3-4  [X]       [2]      [  10](30) [PRS](HGT)
PRES(HGT) TND .1 MB(DM)    0-1-2-3    [ ]       [2]      [   5](2)  [PRT](CHG)
MIXING RAT.     .1G/KG       1-2      [ ]       [2]      [   1]     [MXR]
MAX/MIN TMP     DEG F        1-2      [ ]       [2]      [   5]     [MXM]
ALT. STG.       .01IN HG     1-2      [ ]       [2]      [  10]     [ALT]
DEW POINT(DEP)  DEG F(C)   0-1-2-3    [ ]       [2]      [   5](5)  [DPT](DEP)
PRECIP(ISOTAC)  .01IN(KTS)   1-2-3    [ ]       [2]      [   5](10) [PCP](ISO)
VORTICITY       *10-5S-1   0-1-2-3-4  [X]       [2]      [  20]     [VOR]
CONVERGENCE     *10-5S-1   0-1-2-3    [ ]       [2]      [  10]     [CNV]
MST CONV.       *10-5S-1     1-2      [ ]       [2]      [  10]     [MCV]
******************* FILE AND MAP BACKGROUND INFORMATION *************************
INPUT FILE TYPE [4] (0-PLOT,1-SAODATA,2-OTHER,3-UPPER,4-850MB SAODATA)
INPUT FILE NAME                   [           ] (FOR TYPES 0, 2, AND 3)
OUTPUT GRAPHIC-S MAP BACKGROUND (2,3)[2]   ; SAVE GRIDED DATA (1-YES,0-NO)    [0]
BASE LONGITUDE (LEFT EDGE IN DEG.)    [127] ; X GRID LENGTH (DEGREES) [1.0]
BASE LATITUDE (BOTTOM EDGE IN DEG.)   [31]  ; Y GRID LENGTH (DEGREES) [1.0]    [ ]
```

Figure 2. An example of a preformat menu filled out to request all three 850-mb fields. Recommended contour intervals of 2 for potential temperature, 10 for heights, and 20 for vorticity are used. Input file type 4 is requested in order to generate 850-mb plots and the North American map background is specified.

C. Output:

The programs will produce AFOS graphics with an analysis grid ranging from 98W to 122W and 31N to 51N. Defaults for graphic names are 8MT for temperature, 8MH for height, and 8MV for vorticity. See Figures 3 to 5. Different names for output graphics can be obtained by filling out the desired name in the name box of the preformat.

D. Cautions and Restrictions:

All cautions outlined in the original mesoscale CP apply.

Programming Credits:

MESO2E, GETSAO -- Dave Gilhousen



Figure 3. 850-mb Heights for 22Z, November 5, 1982



Figure 4. 850-mb Potential Temperature Field for 22Z, November 5, 1982



Figure 5. 850-mb Geopotential Vorticity for 22Z, November 5, 1982

E. Complete Program Listing:

```
C***********************************************************************
C
C  PROGRAM MESO2E.FR     GILHOUSEN   1980
C          MODIFIED BY JEFFREY ANDERSON    WRH:SSD    7/82
C
C  MESO2E IS CHAINED TO BY SURFACE IN THE 850MB REDUCTION SERIES.
C  MESO2E USES THE MANALINFO FILE TO DETERMINE GRID BOUNDARIES FOR
C  THE ANALYSIS.  SUBROUTINE GETSAO IS CALLED TO PRODUCE 850MB HEIGHT
C  AND POTENTIAL TEMPERATURE FIELDS.  STATION LATITUDE, LONGITUDE,
C  AND THE TWO DATA FIELDS ARE WRITTEN TO MANALOBS.  MESO2E THEN
C  CHAINS TO MESO3 IN THE MAIN MESOSCALE SERIES.
C
C  VARIABLES AND ARRAYS:
C      INHEADER--        CONTAINS MANALINFO DATA
C      IWXOBS--          ARRAY OF DESIRED DATA FIELDS RETURNED FROM GETSAO
C      LAT,LON--         ARRAYS OF STATION LATITUDE AND LONGITUDE
C      ITIME--           DAY DATE HOUR ARRAY
C      RLONMX,RLONMN-- MAX AND MIN VALUES FOR GRID LONGITUDE
C      RLATMX,RLATMN-- MAX AND MIN GRID LATITUDE VALUES
C      DX,DY--           GRID LENGTH IN TENTHS OF DEGREES
C      BASLON,BASLAT-- COORDINATES OF LOWER LEFT GRID CORNER
C************************************************************************
      PARAMETER RTD=.01745, MAXSTA=400, MAXBASIC=4

      COMMON/HEADR/ INHEADER(46),IWXOBS(MAXSTA,MAXBASIC),
     2             LAT(MAXSTA),LON(MAXSTA)
      COMMON/GRD/ RLONMX,RLONMN,RLATMX,RLATMN,DX,DY,BASLON,BASLAT

      DIMENSION ITIME(4), IWORD(MAXBASIC), JHEADER(3)

      EQUIVALENCE (LOB,INHEADER(28)),(LAB,INHEADER(29)),
     2            (IWORD(1),INHEADER(1)),(NBASIC,INHEADER(17)),
     3            (MAXX,INHEADER(32)),(MAXY,INHEADER(33)),
     4            (LONDEL,INHEADER(30)),(LATDEL,INHEADER(31))

C  GET A CHANNEL AND READ MANALINFO INTO INHEADER
      CALL GCHN(KCHN,IER)
      CALL OPENN(KCHN,'MANALINFO',0,IER)
      CALL RDS(KCHN,INHEADER,92,IER)
      CALL ERROR(IER,'READING MANALINFO')
      CALL KLOSE(KCHN,IER)

C CALCULATE THE REAL BOUNDS OF THE DATA INPUT AREA .
      BASLON=LOB
      DX=LONDEL*0.1
      RLONMX=BASLON+DX
      RLONMN=BASLON-MAXX*DX
      BASLAT=LAB
      DY=LATDEL*0.1
      RLATMX=BASLAT+MAXY*DY
      RLATMN=BASLAT-DY

C  CALL GETSAO TO GET 850MB DATA FIELDS
      CALL GETSAO (IWORD, ITIME, NBASIC, NSTA)

C  PREPARE HEADER FOR MANALOBS FILE
      JHEADER(1)=NSTA
      JHEADER(2)=MAXSTA
      JHEADER(3)=MAXBASIC
      ITIME(4)=ITIME(4)/100

C  DELETE OLD MANALOBS AND OPENN NEW ONE
      CALL DELETE('MANALOBS',IER)
      CALL CREATE('MANALOBS',IER)
```

```
      CALL GCHN(KCHN,IER)
      CALL OPENN(KCHN,'MANALOBS',0,IER)
      CALL ERROR(IER,'ERR MESO2S-OPENNING MANALOBS')

C  WRITE DATE TIME HEADER TO MANALOBS
      CALL WRS(KCHN,ITIME(4),2,IER)
      CALL WRS(KCHN,ITIME(2),2,IER)
      CALL WRS(KCHN,ITIME(1),2,IER)
      CALL WRS(KCHN,ITIME(3),2,IER)

C  WRITE HEADER TO MANALOBS
      CALL WRS(KCHN,JHEADER,6,IER)

C  WRITE LATITUDE AND LONGITUDE TO MANALOBS FILE
      NBYTES=MAXSTA*2
      CALL WRS(KCHN,LON,NBYTES,IER)
      CALL WRS(KCHN,LAT,NBYTES,IER)

C  WRITE DATA FIELDS TO MANALOBS
      NBYTES=MAXSTA*MAXBASIC*2
      CALL WRS(KCHN,IWXOBS,NBYTES,IER)
      CALL KLOSE(KCHN,IER)

C  CHAIN TO MESO3 IN MAIN MESOSCALE SERIES
      CALL FCHAN('MESO3.SV')
      STOP
      END

R
```

```
TYPE GETSAO.FR
C*****************************************************************
C
C  SUBROUTINE GETSAO                        GILHOUSEN 5/1980
C           MODIFIED BY  JEFFREY L. ANDERSON   WRH:SSD  7/82
C
C  GETSAO READS THE FILES REQUESTED BY A MANALINFO FILE FROM
C  AN SAODATA FILE.  FOR AN 850MB PLOT, TEMPERATURE, DEWPOINT, ALTIMETER
C  SETTING AND WIND SPEED ARE READ.  STATION ELEVATION, LATITUDE AND
C  LONGITUDE ARE OBTAINED FROM STDIR.MS BY SUBROUTINE BNSCH. SUBROUTINE
C  H850 PRODUCES 850MB HEIGHT AND POTENTIAL TEMPERATURE FIELDS.  SUBROUTINE
C  BOGUS PRODUCES A NEW FAKE MANALINFO FILE.
C
C       VARIABLES AND ARRAYS:
C          IWORD--        FIRST 4 WORDS OF MANALINFO
C          IC1,IC--       USED IN BNSCH
C          IELEV--        ARRAY OF STATION ELEVATIONS
C          (ALSO SEE VARIABLES DEFINED IN MESO2E)
C*****************************************************************
       PARAMETER LREC=22,ISTAR=24,IFLDP=1,IFLD=6,
     2 MAXSTA=400,MAXBASIC=4, MAXELEV=2060, MINELEV=700

       SUBROUTINE GETSAO(IWORD,IHEADER,NWORDS,NSTA)
       COMMON  /GRD/   RLONMX,RLONMN,RLATMX,RLATMN,DX,DY,BASLON,BASLAT
       COMMON/HEADR/INHEADER(46),IWXOBS(MAXSTA,MAXBASIC),LAT(MAXSTA),
     2 LON(MAXSTA)
       COMMON/CK/ MAX, MIN, PNTRS


       INTEGER MAX(10),MIN(10),PNTRS(10), XCOORD(MAXSTA), YCOORD(MAXSTA)
       DIMENSION IC1(33), IAD(2), IELEV(MAXSTA)
       DIMENSION IWORD(MAXBASIC), IHEADER(4), IBUF(48)

       DATA    MIN/9000,-40,-40,0,0,0,2700,0,0,-40/
       DATA    MAX/10600,125,90,300,360,70,3200,60,500,125/
       DATA    PNTRS/28,29,30,31,32,33,35,42,43,47/

C  OPENN SAODATA FILE
       CALL GCHN(JCHN,IER)
       CALL OPENN (JCHN, 'SAODATA', 0, IER)
       CALL ERROR (IER, 'RDSAO ERROR IN OPENNING CHANNEL')
       CALL GCHN (LCHN, IER)
       CALL OPENN (LCHN, 'STDIR.MS', 0, IER)
       CALL ERROR (IER, 'RDSAO ERROR IN OPENNING CHANNEL')

C  READ 4 HEADER DATE/TIME WORDS
       CALL RDS (LCHN, NREC, 2, IER)
       NBYTES=8
       CALL RDS (JCHN, IHEADER, NBYTES, IER)
C  SET NBYTES TO READ IN ONE STATION RECORD
       NBYTES=96
       NSTA=1

C  LOOP TO READ EACH STATIONS DATA
       DO 200 NS=1,MAXSTA
       CALL RDS(JCHN,IBUF,NBYTES,IER)
       IF(IER.EQ.9) GO TO 250
       IF(IBUF(3).NE.1) GO TO 200      ;CHECK FOR BAD RETURN CODE
       IF(IBUF(48).GT.3) GO TO 200     ;IF NOT PLOTTED, DON'T USE IT!
       IBUF(3)=20040K
C  USE BNSCH TO LOCATE SATION IN STDIR.MS
       CALL BNSCH(LCHN,NREC,LREC,ISTAR,IFLDP,IFLD,IBUF,IAD,IC1,IC1(12),IC1(23),IC)
       IF(IC.EQ.0) GO TO 200
       IC=(IC-1)*11+6
C  GET ELEVATION FROM STDIR AND CHECK FOR LESS THAN MIN OR GREATER THAN MAX
       IELEV(NSTA)=IC1(IC-1)
       IF (IELEV(NSTA).GT.MAXELEV.OR.IELEV(NSTA).LT.MINELEV) GO TO 200
C  GET LATITUDE AND LONGITUDE FROM STDIR.MS
       RLAT=IC1(IC)*0.01
```

```
      RLON=IC1(IC+1)*0.01
C  CHECK FOR DATA OUTSIDE OF GRID
      IF((RLAT.LT.RLATMN) .OR. (RLAT.GT.RLATMX)) GO TO 200
      IF((RLON.LT.RLONMN) .OR. (RLON.GT.RLONMX)) GO TO 200
C  CONVERT DATA TO REAL VALUES IN HUNDREDTHS
      RLAT=((RLAT-BASLAT+DY)/DY)*100
      RLON=((BASLON-RLON+DX)/DX)*100
      LAT(NSTA)=INT(RLAT)
      LON(NSTA)=INT(RLON)
      LSTA=NSTA
      NSTA=NSTA+1
C  READ IN NEEDED DATA FIELDS FOR STATION
      DO 190 NW=1,NWORDS
      DO 160 M=1,10
      IW=M
      IF(IWORD(NW).EQ.PNTRS(M)) GO TO 170
 160  CONTINUE
      GO TO 200
 170  MM=PNTRS(IW)
C  DO GROSS ERROR CHECKS
      IF(IBUF(MM).LT.MIN(IW).OR.IBUF(MM).GT.MAX(IW)) IBUF(MM)=-9999
      IF(MM.NE.42) GO TO 190
      IF(IBUF(41).GT.4) IBUF(MM)=-1*IBUF(MM)
 190  IWXOBS(LSTA,NW)=IBUF(MM)
 200  CONTINUE

 250  NSTA=NSTA-1
      ITIME=IHEADER(4)/100

C  COMPUTE 850MB HEIGHT AND POTENTIAL TEMP
      DO 260 LJI=1, NSTA
 260  CALL H850(IELEV(LJI),IWXOBS(LJI,1),IWXOBS(LJI,2),
     2 IWXOBS(LJI,3),ITIME,IWXOBS(LJI,4))
      CALL KLOSE(JCHN,IER)
      CALL KLOSE(LCHN,IER)

C  GENERATE BOGUS MANALINFO FILE
      CALL BOGUS (INHEADER)

    RETURN
    END
```

```
R
TYPE H850.FR
C*****************************************************************
C   SUBROUTINE H850    JEFFREY L. ANDERSON    WRH:SSD    7/26/82
C
C   SUBROUTINE H850 USES SURFACE ELEVATION, TEMPERATURE, DEWPOINT,
C   ALTIMETER SETTING, WIND VELOCITY AND TIME TO REDUCE SURFACE DATA
C   TO 850MB AND DERIVES 850MB POTENTIAL TEMPERATURE AND HEIGHTS.
C   THE ROUTINE IS PASSED DATA FROM GETSAO AND RETURNS POTENTIAL
C   TEMPERATURE IN ITEMP AND HEIGHT IN IDEW.
C
C   VARIABLES AND ARRAYS:
C        IELEV,ELEV--       ELEVATION
C        ITMP,TEMP--        TEMPERATURE
C        IDEW,DEW--         DEWPOINT
C        IALTIM,ALTIM--     ALTIMETER SETTING
C        ITIME--            HOUR OF OBSERVATION
C        IWIND--            WIND VELOCITY
C        ZEXP--             EXPONENT TO COMPUTE VAPOR PRESSURE
C        VAPOR--            VAPOR PRESSURE
C        PRESS--            STATION PRESSURE
C        NUMER,DENOM--      NUMERATOR AND DENOMINATOR FOR DERIVING 850
C                           HEIGHT
C*****************************************************************
         PARAMETER A1=.005, A2=.0065, ZN=.1902632, C=.12, P=29.921,
     2   T=288.15, ZK=.0148275, MINHGT=1000, MAXHGT=2000

         SUBROUTINE H850 (IELEV, ITEMP, IDEW, IALTIM, ITIME, IWIND)
         REAL NUMER

C   ALL PARAMETERS ARE INTEGER EXCEPT IALTIM WHICH IS 100THS OF AN
C   INCH.
C   H850 IS MISSING IF ANY DATUM IS MISSING
         IF(ITEMP.EQ.-9999 .OR. IDEW.EQ.-9999 .OR. IALTIM.EQ.-9999) GOTO 10
C   CONVERT TO REAL PARAMS.
         ELEV=FLOAT(IELEV)
         TEMP=FLOAT(ITEMP)
C   CONVERT TO KELVIN
         TEMP=(TEMP-32)*5/9+273
         IF(IWIND.GE.14 .AND. ITIME.GT.4 .AND. ITIME.LT.16) GO TO 20
C   USE TIME OF DAY FUNCTION ON T
         TEMP=TEMP-2.9*SIN(2./24.*3.1415*(ITIME-16))
 20      DEW=FLOAT(IDEW)
C   CONVERT TO CELSIUS AND WHOLE INCHES
         DEW=(DEW-32)*5/9
         ALTIM=FLOAT(IALTIM)/100.
C   COMPUTE VAPOR PRESSURE
         ZEXP=7.5*DEW/(DEW+237.3)
         VAPOR=6.11*10.**ZEXP
C   COMPUTE SURFACE PRESSURE
         ARG=ALTIM**ZN-P**ZN*A2/T*ELEV
         PRESS=ARG**(1/ZN)
C   CONVERT PRESSURE FROM INCHES HG TO MB
         PRESS=1013.25/29.921*PRESS
C   COMPUTE NUMERATOR
         NUMER=(TEMP+VAPOR*C)*ALOG10(850./PRESS)
C   COMPUTE DENOMINATOR
         DENOM=ZK-A1/2.*ALOG10(850./PRESS)
C   COMPUTE 850 HEIGHT
         IDEW=IELEV-IFIX((NUMER/DENOM))
C   COMPUTE THETA AT 850 MB
         T850=TEMP-FLOAT(IDEW-IELEV)*.005
         ITEMP=T850*(1000./850.)**.286
C   CHECK FOR BAD HEIGHTS AND RETURN MISSING FOR THESE VALUES
         IF (IDEW.LT.1000 .OR. IDEW.GT.2000) GOTO 10
         RETURN
 10      IDEW = -9999
         ITEMP=-9999
         RETURN
         END
R
```

```
TYPE BOGUS.FR
C*********************************************************************
C  SUBROUTINE BOGUS     JEFFREY L. ANDERSON     WRH:SSD     7/26/82
C
C  BOGUS IS PASSED THE ORIGINAL MANALINFO FILE IN THE ARRAY INHEADER.
C  IT THEN USES THE COUNTOUR INTERVALS FROM THE OLD MANALINFO TO
C  CREATE A NEW BOGUS MANALINFO FILE THAT PRODUCES 850MB PLOTS.
C*********************************************************************
        SUBROUTINE BOGUS (INHEADER)

        COMMON /JB/ IOUT(46)
        COMMON /LOCAL/ MANAL(45)
        DIMENSION INHEADER(46)
        DATA MANAL /29,28,0,0,1,2,0,0,2,2,0,0,3,10,0,0,2,2,2,0,0,0,
     2 '            ',122,31,8,10,24,20,9,'8MT ','8MH ',0,0,0,0,850/
C  DELETE OLD FILES AND OPENN NEW MANALINFO FILE
        CALL GCHN (ICHN,IER)
        CALL DELETE ('MANALINFO',IER)
        CALL CREATE ('MANALINFO',IER)
        CALL OPENN (ICHN,'MANALINFO',0,IER)
        CALL ERROR (IER,'BOGUS ERROR OPENNING')

C  PUT INFO FROM OLD MANALINFO IN NEEDED PLACES
        DO 110 I=9,20
 110    IOUT(I)=INHEADER(I)
        DO 120 I=35,46
 120    IOUT(I)=INHEADER(I)

C  ADD BOGUS PARAMETERS
        DO 10 I=1,8
 10     IOUT(I)=MANAL(I)
        IOUT(17)=2
        DO 40 I=21,34
 40     IOUT(I)=MANAL(I)

C  WRITE MANALINFO FILE
        NBYTES=92
        CALL WRS (ICHN,IOUT,NBYTES,IER)
        CALL KLOSE (ICHN,IER)
        RETURN
        END
R
```

```
C*********************************************************************
C
C  SUBROUTINE MESO850       JEFFREY ANDERSON    WRH:SSD      8/16/82
C
C  MESO850 PRODUCES A MANALINFO FILE FOR 850MB FIELDS DERIVED FROM
C  SURFACE DATA.  SUBROUTINE SAODATA IS CALLED AND ITS MANALINFO FILE
C  IS THEN MODIFIED FOR THE 850MB FORMAT.
C
C  VARIABLES AND ARRAYS:
C        IHEADER--         MANALMENU DATA USED IN SAODATA
C        OUT--             OUTPUT TO MANALINFO
C        IFLD,IFIELD,NB--USED IN SAODATA
C        MANAL,MANA2--     DEFAULT VALUES FOR MANALINFO
C        SMOOTH,CONTOUR,NAME--TEMPORARY INFO ABOUT FIELDS
C*********************************************************************
         SUBROUTINE MESO850 (ICHN)

         COMMON /HE/ IHEADER(128)
         COMMON /PUT/ OUT
         COMMON /ALL/ IFLD(7), IFIELD(14), NB
         COMMON /LD1/ MANA2(2)
         COMMON /LOCAL/ MANAL(43), CONTOUR, SMOOTH, NAME(2,2)
         INTEGER OUT(43), SMOOTH(2), CONTOUR(2)

         DATA NAME /'8MT 8MH '/      ;DEFAULT GRAPHIC NAMES
         DATA MANAL /29,30,35,33,1,9,8,9*0,4,3,0,'8MT 8MH ',4*0,122,31,8,10,24,20,9*0,850/
         DATA CONTOUR/2,10/ SMOOTH/2,2/         ;DEFAULT SMOOTHING AND CONTOURING

C  CALL SAODATA TO LOAD ARRAY OUT WITH VALUES FROM MENU
         CALL SAODATA(ICHN)
C  LOOK FOR TMP AND PRS FIELDS;ERROR ON OTHER FIELDS
         DO 10 I=1,4
         IF (OUT(I).EQ.29) GO TO 20       ;TEMP
         IF (OUT(I).EQ.28) GO TO 30       ;PRESSURE
C  U AND V WIND FIELDS FOR VORTICITY ARE ALSO PERMITTED
         IF (OUT(I).EQ.0 .OR. OUT(I).EQ.32 .OR. OUT(I).EQ.33) GO TO 10
         CALL FORKE ('MESO NOT','ILLEGAL FIELD',IER)       ;ILLEGAL FIELD
10       CONTINUE

C  LOAD DATA DEFAULT VALUES INTO OUPUT FILE FROM MANAL
         DO 50 I=1,17
         OUT(I)=MANAL(I)
 50      CONTINUE
         DO 60 I=23,43
         OUT(I)=MANAL(I)
 60      CONTINUE

C  PUT SMOOTHING PASSES AND CONTOUR INTERVALS FOR TMP
C  AND HGT IN OUTPUT FILE
         DO 70 I=1,2
         OUT(I+8)=SMOOTH(I)
         OUT(I+12)=CONTOUR(I)
 70      CONTINUE
```

```
C  PUT NAMES OF OUTPUT GRAPHICS IN OUTPUT FILE
        DO 80 I=1,4
 80     OUT(34+I)=NAME(I)
C  CHECK FOR NO VORT FIELD AND GET VORT GRAPHIC NAME
        IF (OUT(20).EQ.0) OUT(43)=849      ;NO VORT FIELD REQUIRED
        IF (MANA2(1).NE.'VO' .OR. MANA2(2).NE.'R ') GO TO 90
        MANA2(1)='8M'      ;DEFAULT VORT GRAPHIC NAME
        MANA2(2)='V '

 90     RETURN

C  GET SMOOTHING, CONTOUR INTERVAL AND GRAPHIC NAME FOR TEMP FIELD
 20     SMOOTH(1)=OUT(I+8)
        CONTOUR(1)=OUT(I+12)
        NAME(1,1)=OUT(33+2*I)
        NAME(2,1)=OUT(34+2*I)
        IF (NAME(1,1).EQ.'TM' .AND. NAME(2,1).EQ.'P ') GO TO 110
        GO TO 10

C  GET SMOOTHING, CONTOUR INTERVAL AND GRAPHIC NAME FOR HGT FIELD
 30     SMOOTH(2)=OUT(I+8)
        CONTOUR(2)=OUT(I+12)
        NAME(1,2)=OUT(33+2*I)
        NAME(2,2)=OUT(34+2*I)
        IF (NAME(1,2).EQ.'PR' .AND. NAME(2,2).EQ.'S ') GO TO 120
        GO TO 10

C  TEMP GRAPHIC NAME DEFAULT
 110    NAME(1,1)='8M'
        NAME(2,1)='T '
        GO TO 10

C  HEIGHT GRAPHIC NAME DEFAULT
 120    NAME(1,2)='8M'
        NAME(2,2)='H '
        GO TO 10

        END
```

```
TYPE NEWMESO:BNSCH.FR
      SUBROUTINE BNSCH(ICHN,NREC,LREC,ISTAR,IFLDP,IFLD,ITEST,
     1   IAD,IC1,IC2,IC3,IC)
C     BINARY SEARCH ROUTINE:
C
C     PROGRAMMER - RICH THOMAS SXB,ISL,SDO 11/79
C
C     ICHN=CHANNEL WHICH FILE HAS BEEN OPENNED TO
C     NREC=NUMBER OF RECORDS
C     LREC=LENGTH OF EACH RECORD (BYTES)
C     ISTAR=BYTE OF FIRST RECORD (0=BEGINNING)
C     IFLDP=WORD POINTER TO FIELD IN RECORD
C     IFLD=LENGTH OF FIELD IN BYTES
C     ITEST=ARRAY CONTAINING TEST FIELD
C     IAD=RETURNED TWO WORD ARRAY CONTAINING ADDRESS ITEST RECORD
C       SHOULD BEGIN AT-
C       IC= 1,2,3 IN SECOND WORD INDICATING RECORD WAS FOUND AND
C       IS IN ARRAY IC1,IC2, OR IC3
C     THOSE THREE ARRAYS SHOULD BE DIMENSIONED LREC/2 WORDS
      DIMENSION ITEST(1),IC1(1),IC2(1),IC3(1),IAD(2)
      DIMENSION IAD1(2),IAD2(2),IAD3(2)
      DIMENSION D1(2),D2(2)
      INTEGER D1,D2
      IC=0
      IAD1(1)=0
      IAD1(2)=ISTAR
      CALL SPOS(ICHN,IAD1,IER)
      CALL ERROR(IER,'I1')
      CALL RDS(ICHN,IC1,LREC,IER)
      CALL ERROR(IER,'RDS - IC1')
      D2(1)=0
      D2(2)=LREC
      CALL DSUB(D2,D2,IAD1)
      CALL DMPY(D1,NREC,LREC)
      CALL DSUB(IAD2,D1,D2)
      CALL SPOS(ICHN,IAD2,IER)
      CALL ERROR(IER,'I2')
      CALL RDS(ICHN,IC2,LREC,IER)
      CALL ERROR(IER,'RDS-IC2')
      CALL BCOMP(IC1(IFLDP),ITEST,IFLD,IER1)
      IF(IER1.GT.1)GO TO 100
      CALL BCOMP(IC2(IFLDP),ITEST,IFLD,IER2)
      IF(IER2.NE.2)GO TO 125
    5 CALL DSUB(D1,IAD2,IAD1)
      CALL DDVD(INC,IR,D1,LREC)
      IF(INC.GE.32767)GO TO 900
      IF(INC.LT.1)GO TO 150
      INC=(INC-1)/2+1
      CALL DMPY(D1,INC,LREC)
      CALL DADD(IAD3,IAD1,D1)
      CALL SPOS(ICHN,IAD3,IER)
      CALL ERROR(IER,'I5')
      CALL RDS(ICHN,IC3,LREC,IER)
      CALL ERROR(IER,'I6')
      CALL BCOMP(IC3(IFLDP),ITEST,IFLD,IER3)
      IF(IER3.EQ.1)GO TO 50
      IF(IER3.EQ.2)GO TO 60
      IF(IER3.NE.3)GO TO 900
      IAD(1)=IAD3(1)
      IAD(2)=IAD3(2)
      IC=3
      RETURN
   50 IAD1(1)=IAD3(1)
      IAD1(2)=IAD3(2)
      GO TO 5
   60 IAD2(1)=IAD3(1)
      IAD2(2)=IAD3(2)
      IF(INC.EQ.1)GO TO 150
      GO TO 5
  100 IAD(1)=IAD1(1)
      IAD(2)=IAD1(2)
      IF(IER1.NE.3)GO TO 101
      IC=1
```

```
      IAD(1)=IAD1(1)
      IAD(2)=IAD1(2)
  101 RETURN
  125 D1(1)=0
      D1(2)=LREC
      CALL DADD(IAD,D1,IAD2)
      IF(IER2.NE.3)GO TO 126
      IAD(1)=IAD2(1)
      IAD(2)=IAD2(2)
      IC=2
  126 RETURN
  150 IAD(1)=IAD3(1)
      IAD(2)=IAD3(2)
      RETURN
  900 CALL ERROR(IER3,'IER3')
      IER=2
      CALL ERROR(IER,'TOO MANY RECORDS IN FILE')
      STOP
      END
R
```

```
TYPE MESO4E.FR
C*****************************************************************
C
C  PROGRAM MESO4E    JEFFREY ANDERSON    WRH:SSD    8/18/82
C
C  MESO4E IS CHAINED TO BY MESO4 WHEN AN 850MB VORTICITY FIELD IS
C  NEEDED.  MESO4E READS IN AN 850MB GRIDDED HEIGHT DATA FILE,
C  A NAMES FILE, AND MANALINFO.  U AND V WIND COMPONENTS ARE COMPUTED
C  AS U=-1/F*DH/DY AND V=1/F*DH/DX.  THE COMPONENTS ARE SMOOTHED TWICE
C  AND PUT INTO DATA FILES.  THESE FILES AND NEW BOGUS MANALINFO AND
C  NAMES FILES ARE PASSED TO MESO5.
C
C  VARIABLES AND ARRAYS:
C       INFO--          NAMES FILE
C       MANAL--         BOGUS DATA FOR NEW MANALINFO
C       HEIGHT--        ARRAY OF 850MB HEIGHTS
C       DX--            ARRAY OF X GRID LENGTHS IN KILOMETERS
C       F--             ARRAY OF CORIOLIS PARAMETERS
C       DHDY--          DERIVATIVES OF HEIGHT WITH RESPECT TO Y
C       DHDX--          DERIVATIVES OF HEIGHT WITH RESPECT TO X
C       V--             V WIND COMPONENTS
C       U--             U WIND COMPONENTS
C       DUMMY--         TEMPORARY HOLDING ARRAY FOR CALL TO SMOOTH
C       INHEADER--      OLD MANALINFO FILE
C
C*****************************************************************

      PARAMETER MAXX=24, MAXY=20, MAXXMINUS1=23, MAXYMINUS1=19

      COMMON /AA/ INFO(25), NAME(21), MANAL(46), MNAME(2,3)

      DIMENSION IFILE(10), HEIGHT(MAXX,MAXY), DX(MAXY), F(MAXY),
     2 DHDY(MAXX,MAXY), DHDX(MAXX,MAXY), V(MAXX,MAXY), U(MAXX, MAXY),
     3 INHEADER(46), DUMMY(MAXX,MAXY)

      EQUIVALENCE (LONBASE,INFO(5)), (LATBASE,INFO(6)), (LONDEL,INFO(7)),
     2 (LATDEL,INFO(8))

      DATA NAME /'UUU    VVV            '/
      DATA MANAL/16*0,2,1,2,502,0,0,'                    ',122,31,8,10,
     2 24,20,9,'              ',0,'8MV '/
      DATA MNAME/'UUU.VVV.MXR.'/

C  READ IN NAMES FILE
      CALL GCHN (ICHN,IER)
      CALL OPENN (ICHN, 'NAMES', 0, IER)
      CALL ERROR (IER, 'ERROR, NAMES MISSING')
      N=50
      CALL RDS (ICHN, INFO, N, IER)
      N=20
      CALL RDS (ICHN, IFILE, N, IER)
      CALL KLOSE (ICHN, IER)

C  READ IN GRIDDED HEIGHTS DATA FILE
      IFILE(7)=0
      CALL GCHN (JCHN, IER)
      CALL OPENN (JCHN, IFILE, 0, IER)
      CALL ERROR (IER, 'DATA FILE MISSING')
      N=1920
      CALL RDS (JCHN, HEIGHT, N, IER)
      CALL KLOSE (JCHN, IER)

C  READ IN MANALINFO
      CALL GCHN (KCHN, IER)
      CALL OPENN (KCHN, 'MANALINFO', 0, IER)
      CALL ERROR (IER, 'MESO4E MANALINFO OPENNING')
      N=92
      CALL RDS (KCHN, INHEADER, N, IER)
      CALL KLOSE (KCHN, IER)

C  LOAD DEFAULT VALUES INTO MANALINFO BLOCK
      DO 300 I=1,18
```

```
 300    INHEADER(I)=MANAL(I)
        DO 310 I=21,44
 310    INHEADER(I)=MANAL(I)

C   GET VALUES FOR Y GRID LENGTH DY
        DY=LATDEL*111.2*100.
        ALAT=LATBASE*10-LATDEL

C   COMPUTE X GRID LENGTH AND VALUES FOR F
        DO 30 LJI=1, MAXY
           ALAT=ALAT+LATDEL
           F(LJI)=2*SIN(ALAT*3.1416/1800.)*7.292E-5
 30        DX(LJI)=FLOAT(LONDEL)*111.2*100.*COS(ALAT*3.1416/1800.)

C   FIND DH/DY
        DO 100 J=1, MAXX
           DO 100 I=2, MAXYMINUS1
              I1=I+1
              I2=I-1
 100          DHDY(J,I)=(HEIGHT(J,I+1)-HEIGHT(J,I-1))/(2*DY)

C   FIND DH/DY FOR EDGES
        DO 110 I=1, MAXX
           DHDY(I,1)=(HEIGHT(I,2)-HEIGHT(I,1))/DY
 110       DHDY(I,MAXY)=(HEIGHT(I,MAXY)-HEIGHT(I,MAXY-1))/DY

C   FIND DH/DX FOR INTERIOR POINTS
        DO 120 J=1, MAXY
           DO 120 I=2, MAXXMINUS1
 120       DHDX(I,J)=(HEIGHT(I+1,J)-HEIGHT(I-1,J))/(DX(J)*2)

C   FIND DH/DX FOR EDGES
        DO 130 J=1, MAXY
           DHDX(1,J)=(HEIGHT(2,J)-HEIGHT(1,J))/DX(J)
 130       DHDX(MAXX,J)=(HEIGHT(MAXX,J)-HEIGHT(MAXX-1,J))/DX(J)

C   GET U AND V WIND COMPONENTS
        DO 140 J=1, MAXY
           DO 140 I=1, MAXX
              U(I,J)=-1*DHDY(I,J)/F(J)
              V(I,J)=DHDX(I,J)/F(J)
X       TYPE 'U WIND IS',U(I,J),' VWIND',V(I,J),I,J
 140    CONTINUE

C   SMOOTH WIND COMPONENTS
        IMULT=4
        NUM=2
        CALL SMOOTH (U, MAXX, MAXY, NUM, DUMMY, IMULT)
        CALL SMOOTH (V, MAXX, MAXY, NUM, DUMMY, IMULT)

C   WRITING OUT UUU AND VVV FILES
C   GET CHANNELS AND DELETE ANY OLD FILES
        CALL DFILW ('UUU', IER)
        CALL DFILW ('VVV', IER)
C   CREATE AND OPEN NEW FILES
        IFILE(4)='UU'
        IFILE(5)='U.'
        CALL CFILW (IFILE, 2, IER)
        CALL GCHN (ICHN,IER)
        CALL OPENN (ICHN, IFILE, 0, IER)
        IFILE(4)='VV'
        IFILE(5)='V.'
        CALL CFILW(IFILE,2,IER)
        CALL GCHN (JCHN, IER)
        CALL OPENN (JCHN, IFILE, 0, IER)
C   WRITE FILES AND CLOSE CHANNELS
        NBYTES=MAXX*MAXY*4
        CALL WRS (ICHN, U, NBYTES,IER)
        CALL WRS (JCHN, V, NBYTES, IER)
        CALL KLOSE (ICHN, IER)
        CALL KLOSE (JCHN, IER)
C   DELETE OLD INFORMATION FILES

        CALL  DELETE ('MANALINFO', IER)
```

```
      CALL DELETE ('NAMES',IER)
C  CREATE AND WRITE SECOND NAMES FILE
      CALL CREATE ('NAMES', IER)
      CALL GCHN (ICHN, IER)
      CALL OPENN (ICHN, 'NAMES', 0, IER)
      NBYTES=14
      DO 200 M=1,3
         IFILE(4)=MNAME(1,M)
         IFILE(5)=MNAME(2,M)
         CALL WRS (ICHN, IFILE, NBYTES, IER)
 200  CONTINUE
      CALL KLOSE (ICHN, IER)

C  CREATE AND WRITE NEW MANALINFO
      CALL GCHN (ICHN, IER)
      CALL CREATE ('MANALINFO', IER)
      CALL OPENN (ICHN, 'MANALINFO',0,IER)
      NBYTES=92
      CALL WRS (ICHN, INHEADER, NBYTES, IER)
      CALL KLOSE (ICHN,IER)

C  CHAIN TO MESO5 FOR VORTICITY
X     TYPE  'READY TO CHAIN TO MESO5'
      CALL FCHAN ('MESO5.SV', IER)
      STOP
      END
R
```

```
TYPE NEWMESO:SMOOTH.FR
C
C
C
C
        SUBROUTINE SMOOTH(GP,LIMX,LIMY,NUM,DUMDUM,IMULT)
C
C
        REAL DUMDUM(LIMX,LIMY)
        REAL    GP(LIMX,LIMY)
C
C
C
C
        IMULT=IMULT*2
        IDIV=IMULT+12
C SMOOTH THE INTERNAL POINTS ONLY WITH A SIMPLE LAPLACIAN.
        NX=LIMX-1
        NY=LIMY-1
20      IF(NUM.EQ.0) RETURN                         ; DONE WHEN N=0
        DO 50 J=2,NX
        JJM=J-2
        JJP=J+2
        IF(J.LE.2)JJM=1
        IF(J.GE.NX)JJP=LIMX
        DO 50 K=2,NY
        KKM=K-2
        KKP=K+2
        IF(K.LE.2)KKM=1
        IF(K.GE.NY)KKP=LIMY
        DUMDUM(J,K)=(2.*GP(J-1,K)+2.*GP(J,K-1)+2.*GP(J+1,K)+2.*GP(J,K+1)
     2+GP(JJM,K)+GP(JJP,K)+GP(J,KKP)+GP(J,KKM)+IMULT*GP(J,K))/IDIV
50      CONTINUE
        DO 51 J=2,NY
        DUMDUM(1,J)=(GP(1,J)+GP(2,J)+GP(1,J+1)+GP(1,J-1))/4.0
        DUMDUM(LIMX,J)=(GP(LIMX,J)+GP(NX,J)+GP(LIMX,J+1)+GP(LIMX,J-1))/4.0
 51     CONTINUE
        DO 52 I=2,NX
        DUMDUM(I,1)=(GP(I,1)+GP(I,2)+GP(I+1,1)+GP(I-1,1))/4.0
        DUMDUM(I,LIMY)=(GP(I,LIMY)+GP(I,NY)+GP(I+1,LIMY)+GP(I-1,LIMY))/4.0
 52     CONTINUE
        DUMDUM(1,1)=(GP(1,1)+GP(2,1)+GP(1,2))/3.0
        DUMDUM(1,LIMY)=(GP(1,LIMY)+GP(2,LIMY)+GP(1,NY))/3.0
        DUMDUM(LIMX,1)=(GP(LIMX,1)+GP(LIMX,2)+GP(NX,1))/3.0
        DUMDUM(LIMX,LIMY)=(GP(LIMX,LIMY)+GP(NX,LIMY)+GP(LIMX,NY))/3.0
        DO 53 J=1,LIMY
        DO 53 I=1,LIMX
        GP(I,J)=DUMDUM(I,J)
 53     CONTINUE
        NUM=NUM-1                                   ; DECREMENT THE COUNT.
        GO TO 20
100     CONTINUE
        RETURN
        END
R
```

36 Soaring Forecast Program. David S. Toronto, July 1982.
37 Program to Work Up Climatic Summary Weather Service Forms (F-6, F-52). Peter G. Mueller, August 1982.
38 The Hovmöller Diagram. Pamela A. Hudadoff, September 1982.

NOAA SCIENTIFIC AND TECHNICAL PUBLICATIONS

The National Oceanic and Atmospheric Administration was established as part of the Department of Commerce on October 3, 1970. The mission responsibilities of NOAA are to assess the socioeconomic impact of natural and technological changes in the environment and to monitor and predict the state of the solid Earth, the oceans and their living resources, the atmosphere, and the space environment of the Earth.

The major components of NOAA regularly produce various types of scientific and technical information in the following kinds of publications:

PROFESSIONAL PAPERS — Important definitive research results, major techniques, and special investigations.

CONTRACT AND GRANT REPORTS — Reports prepared by contractors or grantees under NOAA sponsorship.

ATLAS — Presentation of analyzed data generally in the form of maps showing distribution of rainfall, chemical and physical conditions of oceans and atmosphere, distribution of fishes and marine mammals, ionospheric conditions, etc.

TECHNICAL SERVICE PUBLICATIONS — Reports containing data, observations, instructions, etc. A partial listing includes data serials; prediction and outlook periodicals; technical manuals, training papers, planning reports, and information serials; and miscellaneous technical publications.

TECHNICAL REPORTS — Journal quality with extensive details, mathematical developments, or data listings.

TECHNICAL MEMORANDUMS — Reports of preliminary, partial, or negative research or technology results, interim instructions, and the like.

NATIONAL OCEANIC AND ATMOSPHERIC ADMINISTRATION
NOAA
U.S. DEPARTMENT OF COMMERCE

Information on availability of NOAA publications can be obtained from:

ENVIRONMENTAL SCIENCE INFORMATION CENTER (D822)
ENVIRONMENTAL DATA AND INFORMATION SERVICE
NATIONAL OCEANIC AND ATMOSPHERIC ADMINISTRATION
U.S. DEPARTMENT OF COMMERCE

6009 Executive Boulevard
Rockville, MD 20852